We consider the ordinary differential equation model of the metal rolling process defined in K. Galkowski, E. Rogers, W. Paszke and D. H. Owens, *Linear repetitive process control theory applied to a physical example*, Int. J. Appl. Comput. Sci., 13 (2003), 87-99. In order to do that, we firstly define the Ore algebra formed by the derivative D w.r.t. time t and by the shift operator S acting on the discrete variable k which denotes the pass number.

```
> Alg1 := DefineOreAlgebra(diff=[D,t], dual_shift=[S,k], polynom=[t,k],
> comm=[lambda,lambda1,lambda2,M]):
```

The system matrix is defined by:

```
> R1 := evalm([[D^2+lambda/M-(lambda/lambda1)*D^2*S-(lambda/M)*S,
> lambda/(M*lambda2)]]);
```

$$R1 := \left[\begin{array}{cc} \mathrm{D}^2 + \dfrac{\lambda}{M} - \dfrac{\lambda\,\mathrm{D}^2\,S}{\lambda 1} - \dfrac{\lambda\,S}{M} & \dfrac{\lambda}{M\,\lambda 2} \end{array}\right]$$

Then, the system is defined by the following equation

```
> ApplyMatrix(R1, [y(t,k), FM(t,k)], Alg1)[1,1]=0;
```

$$(\lambda\,\mathrm{y}(t,\,k)\,\lambda 1\,\lambda 2 - \lambda\,\mathrm{y}(t,\,k-1)\,\lambda 1\,\lambda 2 + \mathrm{D}_{1,\,1}(y)(t,\,k)\,M\,\lambda 1\,\lambda 2 - \lambda\,\mathrm{D}_{1,\,1}(y)(t,\,k-1)\,M\,\lambda 2 + \lambda\,\mathrm{FM}(t,\,k)\,\lambda 1)/(M\,\lambda 1\,\lambda 2) = 0$$

which corresponds to (4) of the previously quoted paper. Let us check the structural properties of the previous system (e.g., controllability, parametrizability, flatness).

```
> R1_adj := Involution(R1, Alg1):
```

It is known that the *Alg1*-module associated with the matrix *R1* is torsion-free iff the first extension module of the *Alg1*-module associated with *R1_adj* with values in *Alg1* is 0. We compute this extension module by using *Exti*:

```
> Ext1 := Exti(R1_adj, Alg1, 1);
```

$$Ext1 := [\left[\begin{array}{c} 1 \end{array}\right],\ \left[\begin{array}{cc} M\,\lambda\,\mathrm{D}^2\,S\,\lambda 2 - M\,\mathrm{D}^2\,\lambda 1\,\lambda 2 + \lambda\,S\,\lambda 1\,\lambda 2 - \lambda\,\lambda 1\,\lambda 2 & -\lambda\,\lambda 1 \end{array}\right],\ \left[\begin{array}{c} -\lambda\,\lambda 1 \\ M\,\mathrm{D}^2\,\lambda 1\,\lambda 2 - M\,\lambda\,\mathrm{D}^2\,S\,\lambda 2 + \lambda\,\lambda 1\,\lambda 2 - \lambda\,S\,\lambda 1\,\lambda 2 \end{array}\right]]$$

As the first matrix *Ext1*[1] is an identity matrix, we obtain that the *Alg1*-module associated with *R1* is torsion-free, and thus, the corresponding system is controllable and parametrizable. A parametrization is then given by *Ext1*[3]. Equivalently, we can obtain a parametrization of the system by computing *Parametrization*:

```
> evalm([[y(t,k)], [FM(t,k)]])=Parametrization(R1, Alg1);
```

$$\left[\begin{array}{c} \mathrm{y}(t,\,k) \\ \mathrm{FM}(t,\,k) \end{array}\right] = \left[\begin{array}{c} -\lambda\,\lambda 1\,\xi_1(t,\,k) \\ \lambda\,\lambda 1\,\lambda 2\,\xi_1(t,\,k) - \lambda\,\lambda 1\,\lambda 2\,\xi_1(t,\,k-1) - \lambda\,M\,\lambda 2\,\mathrm{D}_{1,\,1}(\xi_1)(t,\,k-1) + M\,\lambda 1\,\lambda 2\,\mathrm{D}_{1,\,1}(\xi_1)(t,\,k) \end{array}\right]$$

Let us check whether or not the metal rolling process is a flat system. In order to do that, we need to test if the corresponding *Alg1*-module is free. In particular, we need to test whether or not the *Alg1*-module associated with *R1* is reflexive.

```
> Ext2 := Exti(R1_adj, Alg1, 2);
```

$$Ext2 := [\left[\begin{array}{c} 1 \end{array}\right], \left[\begin{array}{c} 1 \end{array}\right], \mathrm{SURJ}(1)]$$

As the first matrix *Ext2*[1] is an identity matrix, we obtain that the metal rolling process defines a reflexive, and thus, a projective *Alg1*-module (as we have two independent variables t and k). We note that this last result can be directly checked by verifying that *R1* has full row rank:

```
> SyzygyModule(R1, Alg1);
```

$$\mathrm{INJ}(1)$$

and it admits a right-inverse *S1*

```
> S1 := RightInverse(R1, Alg1);
```

$$S1 := \left[\begin{array}{c} 0 \\ \frac{M\,\lambda 2}{\lambda} \end{array}\right]$$

i.e., we have *R1 S1*=1:

```
> Mult(R1, S1, Alg1);
```

$$\left[\begin{array}{c} 1 \end{array}\right]$$

Finally, the metal rolling process defines a flat system if there exists an injective parametrization of the system, i.e., if there exists a parametrization admitting a left-inverse:

```
> T1 := LeftInverse(Ext1[3], Alg1);
```

$$T1 := \left[\begin{array}{cc} -\frac{1}{\lambda\,\lambda 1} & 0 \end{array}\right]$$

Therefore, the parametrization *Ext1*[3] admits the left-inverse *T1*, i.e., $T1\,Ext1[3] = 1$,

```
> Mult(T1, Ext1[3], Alg1);
```

$$\left[\begin{array}{c} 1 \end{array}\right]$$

which proves that the system is flat. Moreover, a flat output of the system is then given by:

```
> xi[1](t,k)=ApplyMatrix(T1, [y(t,k), FM(t,k)], Alg1)[1,1];
```

$$\xi_1(t,\,k) = -\frac{\mathrm{y}(t,\,k)}{\lambda\,\lambda 1}$$

Using an approximation of the differentiation by a backward difference discretization with sample period T, K. Galkowski, E. Rogers, W. Paszke and D. H. Owens obtain in "*Linear repetitive process control theory applied to a physical example*", Int. J. Appl. Comput. Sci., 13 (2003), 87-99, a difference equation model of the metal rolling process. In order to define such a new system, let us define the Ore algebra *Alg2* of difference-delay operators.

```
> Alg2 := DefineOreAlgebra(dual_shift=[delta,t], dual_shift=[S,k],
> shift_action=[delta,t,T], polynom=[t,k], comm=[lambda,T,lambda1,lambda2,M]):
```

Introducing the following constants of the system

```
> a1 := 2*M/(lambda*T^2+M); a2 := -M/(lambda^2*T+M);
> a3 := lambda*(T^2+M/lambda1)/(lambda*T^2+M);
> a4 := -2*lambda*M/(lambda1*(lambda*T^2+M));
> a5 := lambda*M/(lambda1*(lambda*T^2+M));
> b := -lambda*T^2/(lambda2*(lambda*T^2+M));
```

$$a1 := \frac{2\,M}{\lambda\,T^2 + M}$$

$$a2 := -\frac{M}{\lambda^2\,T + M}$$

$$a3 := \frac{\lambda\,(T^2 + \frac{M}{\lambda 1})}{\lambda\,T^2 + M}$$

$$a4 := -\frac{2\,\lambda\,M}{\lambda 1\,(\lambda\,T^2 + M)}$$

$$a5 := \frac{\lambda\,M}{\lambda 1\,(\lambda\,T^2 + M)}$$

$$b := -\frac{\lambda\,T^2}{\lambda 2\,(\lambda\,T^2 + M)}$$

the system matrix is defined by:

```
> R2 := evalm([[1-a1*delta-a2*delta^2-a3*S-a4*delta*S-a5*delta^2*S, -b]]);
```

$$R2 := \left[1 - \frac{2\,M\,\delta}{\lambda\,T^2 + M} + \frac{M\,\delta^2}{\lambda^2\,T + M} - \frac{\lambda\,(T^2 + \frac{M}{\lambda 1})\,S}{\lambda\,T^2 + M} + \frac{2\,\lambda\,M\,\delta\,S}{\lambda 1\,(\lambda\,T^2 + M)} - \frac{\lambda\,M\,\delta^2\,S}{\lambda 1\,(\lambda\,T^2 + M)},\; \frac{\lambda\,T^2}{\lambda 2\,(\lambda\,T^2 + M)}\right]$$

The system is then defined by the following equation

```
> ApplyMatrix(R2, [y(t,k), FM(t)], Alg2)[1,1]=0;
```

$$\begin{aligned}&(\mathrm{y}(t,\,k)\,\lambda 1\,\lambda 2\,\lambda^3\,T^3 + \mathrm{y}(t,\,k)\,\lambda 1\,\lambda 2\,\lambda\,T^2\,M + \mathrm{y}(t,\,k)\,\lambda 1\,\lambda 2\,M\,\lambda^2\,T + \mathrm{y}(t,\,k)\,\lambda 1\,\lambda 2\,M^2\\&- 2\,M\,\mathrm{y}(t-T,\,k)\,\lambda 1\,\lambda 2\,\lambda^2\,T - 2\,M^2\,\mathrm{y}(t-T,\,k)\,\lambda 1\,\lambda 2 + M\,\mathrm{y}(t-2\,T,\,k)\,\lambda 1\,\lambda 2\,\lambda\,T^2\\&+ M^2\,\mathrm{y}(t-2\,T,\,k)\,\lambda 1\,\lambda 2 - \lambda^3\,\mathrm{y}(t,\,k-1)\,\lambda 2\,T^3\,\lambda 1 - \lambda\,\mathrm{y}(t,\,k-1)\,\lambda 2\,T^2\,\lambda 1\,M\\&- \lambda^3\,\mathrm{y}(t,\,k-1)\,\lambda 2\,M\,T - \lambda\,\mathrm{y}(t,\,k-1)\,\lambda 2\,M^2 + 2\,\lambda^3\,M\,\mathrm{y}(t-T,\,k-1)\,\lambda 2\,T\\&+ 2\,\lambda\,M^2\,\mathrm{y}(t-T,\,k-1)\,\lambda 2 - \lambda^3\,M\,\mathrm{y}(t-2\,T,\,k-1)\,\lambda 2\,T\\&- \lambda\,M^2\,\mathrm{y}(t-2\,T,\,k-1)\,\lambda 2 + \lambda^3\,T^3\,\mathrm{FM}(t)\,\lambda 1 + \lambda\,T^2\,\mathrm{FM}(t)\,\lambda 1\,M) \Big/ ((\lambda\,T^2 + M)\\&(\lambda^2\,T + M)\,\lambda 1\,\lambda 2) = 0\end{aligned}$$

which corresponds to (5) of K. Galkowski, E. Rogers, W. Paszke and D. H. Owens, *Linear repetitive process control theory applied to a physical example*, Int. J. Appl. Comput. Sci., 13 (2003), 87-99.

Let us check the structural properties of this new system. In order to do that, we firstly define:

```
> R2_adj := Involution(R2, Alg2):
```

Then, the properties of the *Alg2*-module associated with *R2* corresponds to the vanishing of the extension modules of the *Alg2*-module associated with *R2_adj* with values in *Alg2*. Let us compute the first one.

```
> Ext1 := Exti(R2_adj, Alg2, 1);
```

$$
\begin{aligned}
&Ext1 := [[\ 1\ ],\\
&\Big[-\lambda 1\,\lambda 2\,M\,\lambda^2\,T+\lambda^3\,T\,M\,\lambda 2\,\delta^2\,S+2\,\lambda^2\,T\,\delta\,\lambda 1\,M\,\lambda 2-\lambda\,T^2\,\delta^2\,\lambda 1\,M\,\lambda 2\\
&+2\,\delta\,\lambda 1\,M^2\,\lambda 2+\lambda\,S\,\delta^2\,M^2\,\lambda 2-\delta^2\,\lambda 1\,M^2\,\lambda 2-\lambda 1\,\lambda 2\,\lambda\,T^2\,M-2\,\lambda^3\,T\,S\,\delta\,M\,\lambda 2\\
&-\lambda 1\,\lambda 2\,M^2-2\,\lambda\,S\,\delta\,M^2\,\lambda 2-\lambda 1\,\lambda 2\,\lambda^3\,T^3+\lambda\,S\,M^2\,\lambda 2+\lambda^3\,T^3\,S\,\lambda 1\,\lambda 2\\
&+\lambda^3\,T\,S\,M\,\lambda 2+\lambda\,T^2\,S\,\lambda 1\,M\,\lambda 2,\ -\lambda^3\,T^3\,\lambda 1-\lambda\,T^2\,\lambda 1\,M\Big],\\
&\left[-\lambda^3\,T^3\,\lambda 1-\lambda\,T^2\,\lambda 1\,M\right]\\
&\Big[-\lambda^3\,T^3\,S\,\lambda 1\,\lambda 2+\lambda 1\,\lambda 2\,\lambda^3\,T^3-\lambda^3\,T\,M\,\lambda 2\,\delta^2\,S-\lambda^3\,T\,S\,M\,\lambda 2+2\,\lambda^3\,T\,S\,\delta\,M\,\lambda 2\\
&-2\,\lambda^2\,T\,\delta\,\lambda 1\,M\,\lambda 2+\lambda 1\,\lambda 2\,M\,\lambda^2\,T-\lambda\,T^2\,S\,\lambda 1\,M\,\lambda 2+\lambda 1\,\lambda 2\,\lambda\,T^2\,M\\
&+\lambda\,T^2\,\delta^2\,\lambda 1\,M\,\lambda 2+2\,\lambda\,S\,\delta\,M^2\,\lambda 2-\lambda\,S\,\delta^2\,M^2\,\lambda 2-\lambda\,S\,M^2\,\lambda 2-2\,\delta\,\lambda 1\,M^2\,\lambda 2\\
&+\lambda 1\,\lambda 2\,M^2+\delta^2\,\lambda 1\,M^2\,\lambda 2\Big]]
\end{aligned}
$$

As the first matrix *Ext1*[1] is an identity matrix, we obtain that the *Alg2*-module associated with *R2* is torsion-free, and thus, controllable and parametrizable. A parametrization of the system is then defined by *Ext1*[3] or, equivalently, we have:

```
> evalm([[y(t,k)], [FM(t,k)]])=Parametrization(R2, Alg2);
```

$$
\begin{aligned}
&\begin{bmatrix} \mathrm{y}(t,\,k)\\ \mathrm{FM}(t,\,k)\end{bmatrix} =\\
&\left[-\lambda^3\,T^3\,\lambda 1\,\xi_1(k)-\lambda\,T^2\,\lambda 1\,M\,\xi_1(k)\right]\\
&\Big[-\lambda 1\,\lambda 2\,\lambda^3\,T^3\,\xi_1(k-1)+\lambda 1\,\lambda 2\,\lambda^3\,T^3\,\xi_1(k)-\lambda 1\,\lambda 2\,M\,\lambda^2\,T\,\xi_1(k)\\
&-\lambda 1\,\lambda 2\,\lambda\,T^2\,M\,\xi_1(k-1)+2\,\lambda 1\,\lambda 2\,\lambda\,T^2\,M\,\xi_1(k)\Big]
\end{aligned}
$$

If we want to test whether or not the system is flat, we then need to check whether or not the second extension module is 0 as we have two independent variables t and k and *R2* has constant coefficients.

```
> Ext2 := Exti(R2_adj, Alg2, 2);
```

$$Ext2 := [[\ 1\ ],\ [\ 1\ ],\ \mathrm{SURJ}(1)]$$

Therefore, we obtain that the *Alg2*-module associated with *R2* is projective, and thus, free by the Quillen-Suslin theorem. This last result can be easily checked by verifying that *R2* has full row rank:

```
> SyzygyModule(R2, Alg2);
```

$$\mathrm{INJ}(1)$$

and it admits a right-inverse *S2*:

```
> S2 := RightInverse(R2, Alg2);
```

$$S2 := \begin{bmatrix} 0\\ \dfrac{\lambda 2\,(\lambda\,T^2+M)}{\lambda\,T^2}\end{bmatrix}$$

In particular, we check that $R2\,S2 = 1$:

```
> Mult(R2, S2, Alg2);
```

$$[\,1\,]$$

As the *Alg2*-module associated with *R2* is free, i.e., the associated control system is flat, then a flat output of the system is obtained as a left-inverse of an injective parametrization of the system defined by *R2*:

```
> xi[1](t,k)=ApplyMatrix(LeftInverse(Ext1[3], Alg2), [y(t,k), FM(t,k)], Alg2)[1,1];
```

$$\xi_1(t,\,k) = -\frac{\mathrm{y}(t,\,k)}{\lambda 1\,\lambda\,T^2\,(\lambda^2\,T + M)}$$

Sampling the previous system at a multiple of T, i.e., $t = p\,T$ where p is a positive integer, a 2-D discrete model of the metal process is obtained in K. Galkowski, E. Rogers, W. Paszke and D. H. Owens, *Linear repetitive process control theory applied to a physical example*, Int. J. Appl. Comput. Sci., 13 (2003), 87-99. In order to define the new model, we firstly need to define the Ore algebra *Alg3* formed by shift operators *Sp* and *Sk*.

```
> Alg3 := DefineOreAlgebra(dual_shift=[Sp,p], dual_shift=[Sk,k],
> polynom=[p,k], comm=[lambda,T,lambda1,lambda2,M]):
```

Then, the system matrix is defined by:

```
> R3 := evalm([[1-a1*Sp-a2*Sp^2-a3*Sk-a4*Sp*Sk-a5*Sp^2*Sk, -b]]);
```

$$R3 := \left[1 - \frac{2\,M\,Sp}{\lambda\,T^2 + M} + \frac{M\,Sp^2}{\lambda^2\,T + M} - \frac{\lambda\,(T^2 + \frac{M}{\lambda 1})\,Sk}{\lambda\,T^2 + M} + \frac{2\,\lambda\,M\,Sp\,Sk}{\lambda 1\,(\lambda\,T^2 + M)} - \frac{\lambda\,M\,Sp^2\,Sk}{\lambda 1\,(\lambda\,T^2 + M)},\ \frac{\lambda\,T^2}{\lambda 2\,(\lambda\,T^2 + M)}\right]$$

Therefore, the system is defined by the following equation:

```
> ApplyMatrix(R3, [y(p,k), FM(p,k)], Alg3)[1,1]=0;
```

$$\begin{aligned}&(\mathrm{y}(p,\,k)\,\lambda 1\,\lambda 2\,\lambda^3\,T^3 + \mathrm{y}(p,\,k)\,\lambda 1\,\lambda 2\,\lambda\,T^2\,M + \mathrm{y}(p,\,k)\,\lambda 1\,\lambda 2\,M\,\lambda^2\,T + \mathrm{y}(p,\,k)\,\lambda 1\,\lambda 2\,M^2\\&+ 2\,\lambda^3\,M\,\mathrm{y}(p-1,\,k-1)\,\lambda 2\,T + 2\,\lambda\,M^2\,\mathrm{y}(p-1,\,k-1)\,\lambda 2\\&- \lambda^3\,M\,\mathrm{y}(p-2,\,k-1)\,\lambda 2\,T - \lambda\,M^2\,\mathrm{y}(p-2,\,k-1)\,\lambda 2 + M\,\mathrm{y}(p-2,\,k)\,\lambda 1\,\lambda 2\,\lambda\,T^2\\&+ M^2\,\mathrm{y}(p-2,\,k)\,\lambda 1\,\lambda 2 - 2\,M\,\mathrm{y}(p-1,\,k)\,\lambda 1\,\lambda 2\,\lambda^2\,T - 2\,M^2\,\mathrm{y}(p-1,\,k)\,\lambda 1\,\lambda 2\\&- \lambda^3\,\mathrm{y}(p,\,k-1)\,\lambda 2\,T^3\,\lambda 1 - \lambda\,\mathrm{y}(p,\,k-1)\,\lambda 2\,T^2\,\lambda 1\,M - \lambda^3\,\mathrm{y}(p,\,k-1)\,\lambda 2\,M\,T\\&- \lambda\,\mathrm{y}(p,\,k-1)\,\lambda 2\,M^2 + \lambda^3\,T^3\,\mathrm{FM}(p,\,k)\,\lambda 1 + \lambda\,T^2\,\mathrm{FM}(p,\,k)\,\lambda 1\,M)\Big/(\lambda 1\\&(\lambda\,T^2 + M)\,(\lambda^2\,T + M)\,\lambda 2) = 0\end{aligned}$$

Let us check the structural properties of this latter model. In order to do that, we introduce the following matrix:

```
> R3_adj := Involution(R3, Alg3):
```

Then, we compute the first extension module of the *Alg3*-module associated with *R3_adj*.

```
> Ext1 := Exti(R3_adj, Alg3, 1);
```

$$Ext1 := [\left[\ 1\ \right], \Big[- \lambda1\,\lambda2\,\lambda\,T^2\,M - \lambda\,T^2\,Sp^2\,\lambda1\,M\,\lambda2 - \lambda1\,\lambda2\,\lambda^3\,T^3 + \lambda\,Sk\,Sp^2\,M^2\,\lambda2 - Sp^2\,\lambda1\,M^2\,\lambda2 - \lambda1\,\lambda2\,M^2 + \lambda^3\,T\,\lambda2\,M\,Sp^2\,Sk + \lambda^3\,T^3\,Sk\,\lambda1\,\lambda2 + \lambda\,T^2\,Sk\,\lambda1\,M\,\lambda2 + \lambda^3\,T\,Sk\,M\,\lambda2 + \lambda\,Sk\,M^2\,\lambda2 - 2\,\lambda^3\,T\,Sk\,Sp\,M\,\lambda2 - 2\,\lambda\,Sk\,Sp\,M^2\,\lambda2 - \lambda1\,\lambda2\,M\,\lambda^2\,T + 2\,\lambda^2\,T\,Sp\,\lambda1\,M\,\lambda2 + 2\,Sp\,\lambda1\,M^2\,\lambda2, -\lambda^3\,T^3\,\lambda1 - \lambda\,T^2\,\lambda1\,M\Big],$$
$$\left[-\lambda^3\,T^3\,\lambda1 - \lambda\,T^2\,\lambda1\,M\right]$$
$$\Big[- \lambda^3\,T^3\,Sk\,\lambda1\,\lambda2 + \lambda1\,\lambda2\,\lambda^3\,T^3 - \lambda^3\,T\,Sk\,M\,\lambda2 + 2\,\lambda^3\,T\,Sk\,Sp\,M\,\lambda2 - \lambda^3\,T\,\lambda2\,M\,Sp^2\,Sk - 2\,\lambda^2\,T\,Sp\,\lambda1\,M\,\lambda2 + \lambda1\,\lambda2\,M\,\lambda^2\,T + \lambda1\,\lambda2\,\lambda\,T^2\,M + \lambda\,T^2\,Sp^2\,\lambda1\,M\,\lambda2 - \lambda\,T^2\,Sk\,\lambda1\,M\,\lambda2 + 2\,\lambda\,Sk\,Sp\,M^2\,\lambda2 - \lambda\,Sk\,M^2\,\lambda2 - \lambda\,Sk\,Sp^2\,M^2\,\lambda2 + \lambda1\,\lambda2\,M^2 + Sp^2\,\lambda1\,M^2\,\lambda2 - 2\,Sp\,\lambda1\,M^2\,\lambda2\Big]]$$

As the first matrix *Ext1* [1] is an identity matrix, we obtain that the system is controllable and parametrizable. A parametrization of the system is then defined by *Ext1* [3]. Equivalently, we have:

```
> evalm([[y(p,k)], [FM(p,k)]])=Parametrization(R3, Alg3);
```

$$\begin{bmatrix} \mathrm{y}(p,\,k) \\ \mathrm{FM}(p,\,k) \end{bmatrix} = \left[-\lambda1\,\lambda\,T^2\,(\lambda^2\,T + M)\,\xi_1(p,\,k)\right]$$
$$\Big[\xi_1(p,\,k)\,\lambda1\,\lambda2\,M^2 + \xi_1(p,\,k)\,\lambda1\,\lambda2\,\lambda^3\,T^3 + \xi_1(p,\,k)\,\lambda1\,\lambda2\,\lambda\,T^2\,M + \xi_1(p,\,k)\,\lambda1\,\lambda2\,M\,\lambda^2\,T - \xi_1(p-2,\,k-1)\,\lambda^3\,T\,M\,\lambda2 - \xi_1(p-2,\,k-1)\,\lambda\,M^2\,\lambda2 + \xi_1(p-2,\,k)\,\lambda1\,\lambda2\,\lambda\,T^2\,M + \xi_1(p-2,\,k)\,\lambda1\,\lambda2\,M^2 + 2\,\xi_1(p-1,\,k-1)\,\lambda\,M^2\,\lambda2 + 2\,\xi_1(p-1,\,k-1)\,\lambda^3\,T\,M\,\lambda2 - 2\,\xi_1(p-1,\,k)\,\lambda1\,\lambda2\,M^2 - 2\,\xi_1(p-1,\,k)\,\lambda1\,\lambda2\,M\,\lambda^2\,T - \xi_1(p,\,k-1)\,\lambda1\,\lambda2\,\lambda\,T^2\,M - \xi_1(p,\,k-1)\,\lambda^3\,T\,M\,\lambda2 - \xi_1(p,\,k-1)\,\lambda\,M^2\,\lambda2 - \xi_1(p,\,k-1)\,\lambda1\,\lambda2\,\lambda^3\,T^3\Big]$$

Moreover, we easily check that the *Alg3*-module associated with the system defined by *R3* is projective as we have:

```
> Ext2 := Exti(R3_adj, Alg3, 2);
```

$$Ext2 := [\left[\ 1\ \right], \left[\ 1\ \right], \mathrm{SURJ}(1)]$$

This last result can also be checked by verifying that *R3* admits a right-inverse *T3*:

```
> T3 := RightInverse(R3, Alg3);
```

$$T3 := \begin{bmatrix} 0 \\ \dfrac{\lambda2\,(\lambda\,T^2 + M)}{\lambda\,T^2} \end{bmatrix}$$

```
> Mult(R3, T3, Alg3);
```

$$\left[\ 1\ \right]$$

Finally, we know that the system defined by *R3* defines a free *Alg3*-module as *R3* has constant coefficients. Let us check whether or not *Ext1*[3] is an injective parametrization, i.e., whether or not *Ext1*[3] admits a left-inverse.

```
> S3 := LeftInverse(Ext1[3], Alg3);
```

$$S3 := \left[\ -\frac{1}{\lambda 1\,\lambda\,T^2\,(\lambda^2\,T+M)} \quad 0\ \right]$$

```
> Mult(S3, Ext1[3], Alg3);
```

$$[\ 1\ ]$$

Therefore, the system is flat and a flat output is then defined by:

```
> xi[1](p,k)=ApplyMatrix(S3, [y(p,k), FM(p,k)], Alg3)[1,1];
```

$$\xi_1(p,\,k) = -\frac{\mathrm{y}(p,\,k)}{\lambda 1\,\lambda\,T^2\,(\lambda^2\,T+M)}$$

Finally, as in K. Galkowski, E. Rogers, W. Paszke and D. H. Owens, *Linear repetitive process control theory applied to a physical example*, Int. J. Appl. Comput. Sci., 13 (2003), 87-99, we consider a first order representation of the system defined by *R3*.

```
> Alg4 := DefineOreAlgebra(shift=[tau,p], dual_shift=[delta,k],
> polynom=[p,k], comm=[M,lambda,T,lambda1,lambda2]):
```

The new set of variables is defined by

$$(y(p-1,k), y(p-2,k), y(p-1,k-1), y(p-2,k-1), FM(p,k), y(p,k))$$

and the matrix of the corresponding system can be proved to be defined by:

```
> R4 := evalm([[tau-a1, -a2, -a4, -a5, -b, -a3*delta],
> [-1, tau, 0, 0, 0, 0],
> [delta, 0, -1, 0, 0, 0],
> [tau, 0, 0, 0, 0, -1],
> [0, delta, 0, -1, 0, 0],
> [0, 0, tau, 0, 0, -delta],
> [0, 0, -1, tau, 0, 0]]);
```

$$R4 := \left[\tau - \frac{2\,M}{\lambda\,T^2+M},\ \frac{M}{\lambda^2\,T+M},\ \frac{2\,\lambda\,M}{\lambda 1\,(\lambda\,T^2+M)},\ -\frac{\lambda\,M}{\lambda 1\,(\lambda\,T^2+M)},\ \frac{\lambda\,T^2}{\lambda 2\,(\lambda\,T^2+M)},\ -\frac{\lambda\,(T^2+\frac{M}{\lambda 1})\,\delta}{\lambda\,T^2+M}\right]$$

$$[-1\,,\,\tau\,,\,0\,,\,0\,,\,0\,,\,0]$$
$$[\delta\,,\,0\,,\,-1\,,\,0\,,\,0\,,\,0]$$
$$[\tau\,,\,0\,,\,0\,,\,0\,,\,0\,,\,-1]$$
$$[0\,,\,\delta\,,\,0\,,\,-1\,,\,0\,,\,0]$$
$$[0\,,\,0\,,\,\tau\,,\,0\,,\,0\,,\,-\delta]$$
$$[0\,,\,0\,,\,-1\,,\,\tau\,,\,0\,,\,0]$$

Using the notations

$$x_1(p,k) = y(p-1,k), x_2(p,k) = y(p-2,k), x_3(p,k) = y(p-1,k-1), x_4(p,k) = y(p-2,k-1),$$

the system equations are then given by:

```
> ApplyMatrix(R4, [seq(x[i](p, k), i=1..4), u(p,k), y(p,k)], Alg4)=
> evalm([seq([0], i=1..8)]);
```

$$\left[-\left(2\,M\,x_1(p,\,k)\,\lambda 1\,\lambda 2\,\lambda^2\,T+2\,M^2\,x_1(p,\,k)\,\lambda 1\,\lambda 2-x_1(p+1,\,k)\,\lambda 1\,\lambda 2\,\lambda^3\,T^3 -x_1(p+1,\,k)\,\lambda 1\,\lambda 2\,\lambda\,T^2\,M-x_1(p+1,\,k)\,\lambda 1\,\lambda 2\,M\,\lambda^2\,T-x_1(p+1,\,k)\,\lambda 1\,\lambda 2\,M^2 -M\,x_2(p,\,k)\,\lambda 1\,\lambda 2\,\lambda\,T^2-M^2\,x_2(p,\,k)\,\lambda 1\,\lambda 2-2\,\lambda^3\,M\,x_3(p,\,k)\,\lambda 2\,T -2\,\lambda\,M^2\,x_3(p,\,k)\,\lambda 2+\lambda^3\,M\,x_4(p,\,k)\,\lambda 2\,T+\lambda\,M^2\,x_4(p,\,k)\,\lambda 2-\lambda^3\,T^3\,\mathrm{u}(p,\,k)\,\lambda 1 -\lambda\,T^2\,\mathrm{u}(p,\,k)\,\lambda 1\,M+\lambda\,\mathrm{y}(p,\,k-1)\,\lambda 2\,M^2+\lambda\,\mathrm{y}(p,\,k-1)\,\lambda 2\,T^2\,\lambda 1\,M +\lambda^3\,\mathrm{y}(p,\,k-1)\,\lambda 2\,M\,T+\lambda^3\,\mathrm{y}(p,\,k-1)\,\lambda 2\,T^3\,\lambda 1\right)\Big/\left((\lambda\,T^2+M)\,(\lambda^2\,T+M)\,\lambda 1\,\lambda 2\right)\right]$$

$$\begin{array}{l} [-x_1(p,\,k)+x_2(p+1,\,k)] \\ [x_1(p,\,k-1)-x_3(p,\,k)] \\ [x_1(p+1,\,k)-\mathrm{y}(p,\,k)] \\ [x_2(p,\,k-1)-x_4(p,\,k)] \\ [x_3(p+1,\,k)-\mathrm{y}(p,\,k-1)] \\ [-x_3(p,\,k)+x_4(p+1,\,k)] \end{array} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

We point out that system (6) considered in K. Galkowski, E. Rogers, W. Paszke and D. H. Owens, *Linear repetitive process control theory applied to a physical example*, Int. J. Appl. Comput. Sci., 13 (2003), 87-99, is not algebraically equivalent to the system defined by *R3*. For instance, system (6) is not controllable, and thus, not flat, whereas we have seen that the system associated with *R3* was controllable and flat. Such a problem comes from the fact that the modules defined by *R3* and *R4* are isomorphic, which is not the case with system (6) as some equations are missing in the repetitive process model (6).

Let us study the structural properties of the new system defined by *R4*. We firstly introduced the following matrix:

```
> R4_adj := Involution(R4, Alg4):
```

As it was said previously, the controllability of the system defined by *R4* is related to the vanishing of the first extension module with values in *Alg4* of the *Alg4*-module associated with *R4_adj*. Let us compute it.

```
> Ext1 := Exti(R4_adj, Alg4, 1);
```

$$Ext1 := \left[\begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}\right.,$$

$$\Big[2\,\lambda 1\,\lambda 2\,M^2 + 2\,\lambda 1\,\lambda 2\,M\,\lambda^2\,T\,,\ -\%2 - \%1\,,\ -2\,\lambda\,M^2\,\lambda 2 - 2\,\lambda^3\,T\,M\,\lambda 2\,,$$
$$\lambda\,M^2\,\lambda 2 + \lambda^3\,T\,M\,\lambda 2\,,\ \%3\,, -\lambda 1\,\lambda 2\,\lambda^3\,T^3 - \%1 - \lambda 1\,\lambda 2\,M\,\lambda^2\,T - \%2$$
$$+ M\,\lambda\,T^2\,\delta\,\lambda 1\,\lambda 2 + \lambda\,\delta\,\lambda 2\,M^2 + \lambda^3\,T^3\,\delta\,\lambda 1\,\lambda 2 + \lambda^3\,T\,\delta\,\lambda 2\,M\Big]$$

$$[\delta\,,\ 0\,,\ -1\,,\ 0\,,\ 0\,,\ 0]$$

$$\Big[2\,\lambda 1\,\lambda 2\,M^2 + 2\,\lambda 1\,\lambda 2\,M\,\lambda^2\,T\,,\ -\%2 - \%1\,, \lambda 1\,\lambda 2\,\tau\,T^3\,\lambda^3 + \lambda 2\,\lambda^3\,\tau\,T\,M$$
$$+ \lambda 1\,\lambda 2\,\tau\,T^2\,M\,\lambda + \lambda 2\,\lambda\,\tau\,M^2 - 2\,\lambda\,M^2\,\lambda 2 - 2\,\lambda^3\,T\,M\,\lambda 2,\ \lambda\,M^2\,\lambda 2 + \lambda^3\,T\,M\,\lambda 2\,,$$
$$\%3\,,\ -\%2 - \lambda 1\,\lambda 2\,M\,\lambda^2\,T - \lambda 1\,\lambda 2\,\lambda^3\,T^3 - \%1\Big]$$

$$[-1\,,\ \tau\,,\ 0\,,\ 0\,,\ 0\,,\ 0]$$
$$[\tau\,,\ 0\,,\ 0\,,\ 0\,,\ 0\,,\ -1]$$
$$[0\,,\ \delta\,,\ 0\,,\ -1\,,\ 0\,,\ 0]$$
$$[0\,,\ 0\,,\ -1\,,\ \tau\,,\ 0\,,\ 0],$$
$$[-\tau\,M\,\lambda\,T^2\,\lambda 1 - \tau\,\lambda^3\,T^3\,\lambda 1]$$
$$[\%3]$$
$$[-\tau\,M\,\lambda\,T^2\,\delta\,\lambda 1 - \tau\,\lambda^3\,T^3\,\delta\,\lambda 1]$$
$$[-\delta\,\lambda\,M\,\lambda 1\,T^2 - \delta\,\lambda^3\,T^3\,\lambda 1]$$

$$\Big[\%2 + 2\,\lambda^3\,T\,M\,\lambda 2\,\delta\,\tau - \lambda^3\,T\,\delta\,\lambda 2\,M - 2\,\lambda^2\,T\,\lambda 1\,\tau\,M\,\lambda 2 + 2\,\lambda\,M^2\,\lambda 2\,\delta\,\tau$$
$$- \lambda\,\delta\,\lambda 2\,M^2 - 2\,\lambda 1\,\tau\,M^2\,\lambda 2 + \lambda^3\,T^3\,\lambda 1\,\tau^2\,\lambda 2 - \lambda^3\,T^3\,\tau^2\,\delta\,\lambda 1\,\lambda 2 - \lambda^3\,T\,M\,\lambda 2\,\delta\,\tau^2$$
$$+ \lambda^2\,T\,\lambda 1\,\tau^2\,M\,\lambda 2 + M\,\lambda\,T^2\,\lambda 1\,\tau^2\,\lambda 2 - M\,\lambda\,T^2\,\tau^2\,\delta\,\lambda 1\,\lambda 2 - \lambda\,M^2\,\lambda 2\,\delta\,\tau^2$$
$$+ \lambda 1\,\tau^2\,M^2\,\lambda 2 + \%1\Big]$$

$$\left.[-\lambda^3\,T^3\,\tau^2\,\lambda 1 - M\,\lambda\,T^2\,\tau^2\,\lambda 1]\right]$$

$$\%1 := \lambda 1\,\lambda 2\,\lambda\,T^2\,M$$
$$\%2 := \lambda 1\,\lambda 2\,M^2$$
$$\%3 := -\lambda^3\,T^3\,\lambda 1 - \lambda\,T^2\,\lambda 1\,M$$

As the first matrix *Ext1*[1] is an identity matrix, we obtain that the system associated with *R4* is controllable and parametrizable. A parametrization is then given by *Ext1*[3] or, equivalently, by:

```
> evalm([seq([x[i](p,k)], i=1..4), [u(p,k)], [y(p,k)]])=Parametrization(R4, Alg4);
```

$$\begin{bmatrix} x_1(p,\,k) \\ x_2(p,\,k) \\ x_3(p,\,k) \\ x_4(p,\,k) \\ \mathrm{u}(p,\,k) \\ \mathrm{y}(p,\,k) \end{bmatrix} =$$

$$\left[-\lambda\,T^2\,\lambda 1\,(\lambda^2\,T+M)\,\xi_1(p+1,\,k)\right]$$
$$\left[-\lambda\,T^2\,\lambda 1\,(\lambda^2\,T+M)\,\xi_1(p,\,k)\right]$$
$$\left[-\lambda\,T^2\,\lambda 1\,(\lambda^2\,T+M)\,\xi_1(p+1,\,k-1)\right]$$
$$\left[-\lambda\,T^2\,\lambda 1\,(\lambda^2\,T+M)\,\xi_1(p,\,k-1)\right]$$
$$\Big[\xi_1(p,\,k)\,\lambda 1\,\lambda 2\,M^2+\xi_1(p,\,k)\,\lambda 1\,\lambda 2\,\lambda\,T^2\,M+\xi_1(p+2,\,k)\,\lambda 1\,\lambda 2\,\lambda^3\,T^3$$
$$+\,\xi_1(p+2,\,k)\,\lambda 1\,\lambda 2\,\lambda\,T^2\,M+\xi_1(p+2,\,k)\,\lambda 1\,\lambda 2\,M\,\lambda^2\,T+\xi_1(p+2,\,k)\,\lambda 1\,\lambda 2\,M^2$$
$$-\,\xi_1(p,\,k-1)\,\lambda^3\,T\,M\,\lambda 2-\xi_1(p,\,k-1)\,\lambda\,M^2\,\lambda 2+2\,\xi_1(p+1,\,k-1)\,\lambda^3\,T\,M\,\lambda 2$$
$$+\,2\,\xi_1(p+1,\,k-1)\,\lambda\,M^2\,\lambda 2-2\,\xi_1(p+1,\,k)\,\lambda 1\,\lambda 2\,M\,\lambda^2\,T$$
$$-\,2\,\xi_1(p+1,\,k)\,\lambda 1\,\lambda 2\,M^2-\%1\,\lambda 1\,\lambda 2\,\lambda^3\,T^3-\%1\,\lambda^3\,T\,M\,\lambda 2-\%1\,\lambda 1\,\lambda 2\,\lambda\,T^2\,M$$
$$-\,\%1\,\lambda\,M^2\,\lambda 2\Big]$$
$$\left[-\lambda\,T^2\,\lambda 1\,(\lambda^2\,T+M)\,\xi_1(p+2,\,k)\right]$$
$$\%1 := \xi_1(p+2,\,k-1)$$

Let us check whether or not the system associated with *R4* is flat. In order to do that, we can check the vanishing of the second extension module with value in *Alg4* of the *Alg4*-module associated with *R4_adj*.

```
>  Ext2 := Exti(R4_adj, Alg4, 2);
```

$$Ext2 := [\left[\ 1\ \right],\ \left[\ 1\ \right],\ \mathrm{SURJ}(1)]$$

As the first matrix *Ext2*[1] is an identity matrix, we obtain that the system defined by *R4* generates a free *Alg4*-module, and thus, is flat. A flat output of the system is obtained by computing a left-inverse *T4* of the parametrization *Ext1*[3] of the system.

```
>  T4 := LeftInverse(Ext1[3], Alg4);
```

$$T4 := \left[\ 0 \quad -\frac{1}{\lambda 1\,\lambda\,T^2\,(\lambda^2\,T+M)} \quad 0 \quad 0 \quad 0 \quad 0\ \right]$$

Therefore, a flat output of the system is defined by:

```
>  xi[1](p,k)=ApplyMatrix(T4, [seq([x[i](p,k)], i=1..4), [u(p,k)],
>  [y(p,k)]], Alg4)[1,1];
```

$$\xi_1(p,\,k) = -\frac{x_2(p,\,k)}{\lambda 1\,\lambda\,T^2\,(\lambda^2\,T+M)}$$